
Life table method 與 Kaplan-Meier method 的不同之處?        劉嘉樺 

    存活分析是研究某個群體在某特定時間起經過一段時間的觀察追蹤，直到特定事件的發生，即探

討事件發生所需的時間(time to event)。 

    存活分析中常用的分析方法若按照使用參數與否可分爲三類，參數法、半參數法及非參數法。若

是樣本適合某種分配，則可採用參數法，常用的參數方法有指數分配、Weibull 分配和 Gomperzt

分配模式；半參數方法指的是 Cox 比例風險模型；非參數法表示當被研究事件沒有很好的參數模式

可以配適時，通常可以採用非參數方法研究它的生存特徵，常用的方法包括 Life-table method 和 

Kaplan-Meier method。 

Life-table method 與 Kaplan-Meier method 有何差異？Life-table method 常用來表示存活

情形，多用固定間隔來設定區間，例如：以每一年或每一月為固定間隔。Kaplan-Meier method 又

可稱 Product-limit method，則是用來估計存活曲線，以每一個事件發生時間點與設限點(censoring)

來設定區間，較適用於一般臨床相關研究。在蒐集存活資料時，有時無法確切的記錄所感興趣事件所

發生的時間，此現象稱之為設限(censoring)，造成設限的原因可能是在進行資料分析時所關切的事

件尚未發生；也可能因受測者失去聯絡或中途離開研究。 

在 SAS 中，計算存活函數的語法如下： 

 

PROC LIFETEST 簡單的用法是用非參數法估計樣本存活時間的生存函數。在這種情況下，僅需要

PROC LIFETEST 和 TIME。PROC LIFETEST 語法中，除 TIME 外的所有語法都是可選的，且 PROC 

LIFETEST 之後的語法沒有要求的順序。主要指令說明如下： 

 STRATA：指定要分層的變項。 

 TIME：指定時間變項乘以事件變項(設限符號)。 

 TEST：指定共變數。 

在 PROC LIFETEST 語法中，可用的選項設定還有很多，Life-table method 與 Kaplan-Meier method

該如何設定呢？利用非參數估計中的 METHOD 來指定計算存活函數的方式，METHOD=LT 指



Life-table 的估計、METHOD=KM 指 Kaplan-Meier 的估計。若在沒有設定 METHOD 的情況下，

一般 SAS 會內建為 METHOD=KM。 

SAS 範例 72.3 中，患有心絞痛的男性生命表估計 

本例中的數據來自 Lee（1992，p91），代表患有心絞痛的男性的生存率，生存時間以診斷開始的年

數來計算，共分成 16 個時間間格。根據數據構造三個變數：Years、

Censored(1 表示 censored，0 表示 event)與 Freq。 

title 'Survival of Males with Angina Pectoris'; 

data Males; 

   keep Freq Years Censored; 

   retain Years ‐.5; 

   input fail withdraw @@; 

   Years + 1; 

   Censored=0; 

   Freq=fail; 

   output; 

   Censored=1; 

   Freq=withdraw;   

   output; 

   datalines; 

456   0 226  39 152  22 171  23 135 24 125 107 

 83 133  74 102  51  68  42  64  43 45  34  53 

 18  33   9  27   6  23   0  30 

; 

ods graphics on; 

proc lifetest data=Males  method=lt intervals=(0 to 15 by 1) 

              plots=(s,ls,lls,h,p); 

   time Years*Censored(1); 

   freq Freq; 

run; 

   ods graphics off; 

重要指令說明： 

 method=lt：指定 Life-table method 為計算存活函數的方式。 

 plots=(s,ls,lls,h,p)：依指定函數和時間繪製圖形，可用 PLOTS=ALL 代替。在 Life-table method

中 PLOTS=ALL 等同於 PLOTS=(SURVIVAL LOGSURV LOGLOGS DENSITY HAZARD)；在

Kaplan-Meier method 中 PLOTS=ALL 等同於 PLOTS=(SURVIVAL LOGSURV LOGLOGLS 

HAZARD)。 

 intervals=(0 to 15 by 1)：在 Life-table method 中，指定時間的端點，都必須是非負數。 

 



 

 

Life Table 估計的結果顯示在上表中。五年存活率為 0.5193、標準誤差為 0.0103，估計的平均剩餘

壽命 初為 5.33 年，在第二年開始時達到 大值 6.34 年，然後逐漸降低，其值到第七年低於 初始

值的 5.33 年。 



 

由上表中，可看出事件人數與設限人數，且有 32.8%的患者退出了研究。 

 

上圖為 Life Table 存活函數估計值的圖表，從存活曲線中可知存活時間的中位數為 5.33 年，第 25 個

百分位數和第 75 個百分位數分別為 1.04 年和 11.13 年。存活時間的中位數為 5.33 年，表示在確診

5.33 年後，仍有一半（50%）的患者存活著。 

 

  



上述範例 72.3 是利用 Life-table method 來估計存活曲線，接下來的範例 72.1 是利用 Kaplan-Meier 

method 來估計存活曲線。存活曲線通常是用階梯狀的線條連接，但在 SAS 的 Life-table method

中，存活曲線的預設是用折線來連接不同時間點的存活率。 

SAS 範例 72.1 中，Product-Limit 估計和相關檢定 

本例的數據來自 Kalbfleisch 和 Prentice（1980）附錄 I 中。依變數 SurvTime 代表肺癌患者的存活

天數，SurvTime 的負值是設限值。共變數為 Cell（癌細胞類型）、Therapy（治療類型：standard or 

test）、Prior（早期治療：0=no, 10=yes）、Age（以年為單位）、DiagTime（從診斷到進入試驗的

月數）和 Kps（日常體能狀態）。根據數值建立設限指標變數 Censor，1 表示設限時間，0 表示事件

時間。由於只有兩種類型的治療，因此針對 Therapy(治療類型)建立指標變數 Treatment，0 表示

standard(標準治療)，1 表示 test(測試治療)。 

data VALung; 

   drop check m; 

   retain Therapy Cell; 

   infile cards column=column; 

   length Check $ 1; 

   label SurvTime='Failure or Censoring Time' 

      Kps='Karnofsky Index' 

      DiagTime='Months till Randomization' 

      Age='Age in Years' 

      Prior='Prior Treatment?' 

      Cell='Cell Type' 

      Therapy='Type of Treatment' 

      Treatment='Treatment Indicator'; 

   M=Column; 

   input Check $ @@; 

   if M>Column then M=1; 

   if Check='s'|Check='t' then input @M Therapy $ Cell $ ; 

   else input @M SurvTime Kps DiagTime Age Prior @@; 

   if SurvTime > .; 

   censor=(SurvTime<0); 

   SurvTime=abs(SurvTime); 

   Treatment=(Therapy='test'); 

   datalines; 

standard squamous 

 72 60  7 69  0   411 70  5 64 10   228 60  3 38  0   126 60  9 63 10 

118 70 11 65 10    10 20  5 49  0    82 40 10 69 10   110 80 29 68  0 

314 50 18 43  0  ‐100 70  6 70  0    42 60  4 81  0     8 40 58 63 10 

144 30  4 63  0   ‐25 80  9 52 10    11 70 11 48 10 

standard small 

 30 60  3 61  0   384 60  9 42  0     4 40  2 35  0    54 80  4 63 10 



 13 60  4 56  0  ‐123 40  3 55  0   ‐97 60  5 67  0   153 60 14 63 10 

 59 30  2 65  0   117 80  3 46  0    16 30  4 53 10   151 50 12 69  0 

 22 60  4 68  0    56 80 12 43 10    21 40  2 55 10    18 20 15 42  0 

139 80  2 64  0    20 30  5 65  0    31 75  3 65  0    52 70  2 55  0 

287 60 25 66 10    18 30  4 60  0    51 60  1 67  0   122 80 28 53  0 

 27 60  8 62  0    54 70  1 67  0     7 50  7 72  0    63 50 11 48  0 

392 40  4 68  0    10 40 23 67 10 

standard adeno 

  8 20 19 61 10    92 70 10 60  0    35 40  6 62  0   117 80  2 38  0 

132 80  5 50  0    12 50  4 63 10   162 80  5 64  0     3 30  3 43  0 

 95 80  4 34  0 

standard large 

177 50 16 66 10   162 80  5 62  0   216 50 15 52  0   553 70  2 47  0 

278 60 12 63  0    12 40 12 68 10   260 80  5 45  0   200 80 12 41 10 

156 70  2 66  0  ‐182 90  2 62  0   143 90  8 60  0   105 80 11 66  0 

103 80  5 38  0   250 70  8 53 10   100 60 13 37 10 

test squamous 

999 90 12 54 10   112 80  6 60  0   ‐87 80  3 48  0  ‐231 50  8 52 10 

242 50  1 70  0   991 70  7 50 10   111 70  3 62  0     1 20 21 65 10 

587 60  3 58  0   389 90  2 62  0    33 30  6 64  0    25 20 36 63  0 

357 70 13 58  0   467 90  2 64  0   201 80 28 52 10     1 50  7 35  0 

 30 70 11 63  0    44 60 13 70 10   283 90  2 51  0    15 50 13 40 10 

test small 

 25 30  2 69  0  ‐103 70 22 36 10    21 20  4 71  0    13 30  2 62  0 

 87 60  2 60  0     2 40 36 44 10    20 30  9 54 10     7 20 11 66  0 

 24 60  8 49  0    99 70  3 72  0     8 80  2 68  0    99 85  4 62  0 

 61 70  2 71  0    25 70  2 70  0    95 70  1 61  0    80 50 17 71  0 

 51 30 87 59 10    29 40  8 67  0 

test adeno 

 24 40  2 60  0    18 40  5 69 10   ‐83 99  3 57  0    31 80  3 39  0 

 51 60  5 62  0    90 60 22 50 10    52 60  3 43  0    73 60  3 70  0 

  8 50  5 66  0    36 70  8 61  0    48 10  4 81  0     7 40  4 58  0 

140 70  3 63  0   186 90  3 60  0    84 80  4 62 10    19 50 10 42  0 

 45 40  3 69  0    80 40  4 63  0 

test large 

 52 60  4 45  0   164 70 15 68 10    19 30  4 39 10    53 60 12 66  0 

 15 30  5 63  0    43 60 11 49 10   340 80 10 64 10   133 75  1 65  0 

111 60  5 64  0   231 70 18 67 10   378 80  4 65  0    49 30  3 37  0 

; 

ods graphics on; 

proc lifetest data=VALung plots=(s(test),ls,lls) outtest=Test maxtime=600; 

   time SurvTime*Censor(1); 

   id Therapy; 

   strata Cell; 

   test Age Prior DiagTime Kps Treatment; 

run; 

ods graphics off; 

 



 

……. 

 

重要指令說明： 

 利用 PROC LIFETEST 計算對於每種類型的癌細胞存活函數的 Product-limit 估計值，並分析變

數 Age、Prior、DiagTime、Kps 和 Treatment 對患者存活率的影響。這些預後因素在 TEST 語

句中指定，變數 Cell 在 STRATA 語句中指定。 

 plots=(s(test),ls,lls)：製作圖形之前必須先啟用 ODS 圖形，利用 PROC LIFETEST 語句中的 PLOTS 

= option 來繪製所需圖形。the product-limit estimates (S)、the negative log estimates (LS)、

the log of negative log estimates (LLS)。(test)指令可讓存活函數 log-rank 檢定的 p-value。

顯示在圖形上。 

 maxtime=600：由於生存時間較長，因此使用 MAXTIME 600 來設置時間軸的刻度。也就是說，



在圖中，時間範圍從 0 擴展到 多 600 天。 

 id Therapy：ID 指定了變數 Therapy，以 Product-limit 估計每種觀察值的治療類型。 

 outtest=Test：OUTTEST 選項指定檔名 Test 為輸出數據。 

 

 

 

 

 

 



 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 



 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

上述四個表為利用 Product-limit 估計四種細胞類型的存活函數，還有一些存活時間的摘要統計。對

於腺細胞、大細胞、小細胞和鱗狀細胞患者，存活時間的中位數分別為 51 天，156 天，51 天和 118

天。 



 

上表總結了四種細胞類型中，事件和設限觀測值的分佈。 

 

 

上圖中呈現的是估計存活函數的圖形，腺細胞曲線和小細胞曲線比大細胞曲線或鱗狀細胞曲線彼此更

接近。腺細胞患者和小細胞患者的存活率在 90 天內迅速下降到大約 29％。大細胞曲線和鱗狀細胞曲

線的形狀差異很大，儘管兩者的下降速度均不及腺細胞和小細胞。鱗狀細胞曲線 初比大細胞曲線下

降得更快，但在後期則相反。 



 

上表為細胞類型的同質性檢定結果，表格中包含 rank 統計值(Log-rank 與 Wilcoxon)、Log-rank 統

計值的共變異數矩陣、Wilcoxon 統計值的共變異數矩陣與分層相等性的檢定。分層相等性的檢定中，

提供了三種檢定方式的卡方統計量、自由度與 p-value。以上三種檢定方式均有強力證據表明：四種

癌細胞的存活曲線之間存在顯著的差異（p <0.0001）。 

